HRP 262 SAS LAB FIVE, May 13, 2009

Lab Five: Profile Plots, Repeated-Measures ANOVA

Lab Objectives

After today’s lab you should be able to:

1. Convert data from broad to long form.

2. Fill in missing data with LAST OBSERVATION CARRIED FORWARD (LOCF).

3. Plot individual and mean profile plots using PROC GPLOT.

4. Understand SAS MACROs.

5. Store and call global variables.

6. Write a SAS MACRO for plotting (for future use).

7. Check compound symmetry assumption of univariate repeated-measures ANOVA.

8. Generate correlation and variance-covariance matrices for repeated measurements using PROC CORR.

9. Use ANCOVA for end-point analysis in PROC GLM (broad form of data).

10. Run repeated-measures ANOVA (univariate and multivariate) with PROC GLM (long form of the data).

11. Interpret results from repeated measures ANOVA.

12. Use PROC RANK to categorize a continuous variable into tertiles.

13. Add additional features to make graphs fancier in PROC GPLOT.

 LAB EXERCISE STEPS:

Follow along with the computer in front…

1. For today’s class, we will be using the Lab 4 data (runners.sas7bdat). If this dataset is not already on your desktop( then goto: www.stanford.edu/~kcobb/courses/hrp262 and download the Lab 4-8 data.
2. Open SAS: Start Menu( All Programs(SAS
Name a library using hrp262 that points to the desktop, using point-and-click.

3. Use Last Observation Carried Forward to fill in missing data on our repeated measurement of interest. 
data locf;

set hrp262.runners;

if bmc2=. then bmc2=bmc1;

if bmc3=. then bmc3=bmc2;

run;

4. Copy the locf dataset into a “broad” set with extraneous variables removed. 

Add a unique id number for each subject (_n_ is an automatically generated SAS variable that is the row number of the observation—which totally depends on how the data are sorted). 

data broad;

retain id mencat1 stressf sitenum calc1 treatr dxa bmc;

set locf;

if bmc2=. then bmc2=bmc1;

if bmc3=. then bmc3=bmc2;

id=_n_; *create a unique id number to identify each subject;
keep id mencat1 stressf sitenum calc1 treatr bmc1 bmc2 bmc3;

run;


5. Then reformat the dataset into a “long” form (we will use both the broad and long forms of the data for this lab). 

data long;

set broad;

dxa=1; bmc=bmc1; output;

dxa=2; bmc=bmc2; output;

dxa=3; bmc=bmc3; output;

drop bmc1 bmc2 bmc3;

run;
6. Use Solutions(Analysis(Interactive data analysis to view the broad and long forms of the data.
7. Next, plot BMC over time by individual:


proc sort data=long;

  by id dxa; 

run;

goptions reset=all ftext=oldeng htext=3; *change font type and size (Old English);
proc gplot data=long;

   symbol1 c=black i=join v=none height=2 repeat=78;

  plot bmc*dxa=id/nolegend;

run; quit;

8. It’s hard to make many conclusions with 78 lines cluttering the plot, but it appears that change over time is fairly static.  But let’s fancy the graph up a little bit by overlaying a mean line to view the trends…

/*First, calculate the means at dxa1, dxa2, and dxa3*/
proc means data=long noprint nway; 

  class dxa;

  var bmc;

  output out=means mean=;

run;


proc print data=means;

run;

  


     Obs    dxa    _TYPE_    _FREQ_      bmc

                            1      1        1        78      2178.55

                            2      2        1        78      2190.06

                            3      3        1        78      2197.21
/*Second, tack these means onto the long dataset*/
 data both;

  set means long;

  if _freq_>0 then id=1000; *only means have a _freq_ value;
  keep dxa bmc id _freq_;

  if _n_=1 then call symput("therepeat",_freq_); 

run; 


proc sort data=both;

  by id dxa; 

run; 

goptions reset=all ftext=oldeng htext=3; *change font type and size;
title 'BMC changes over time with mean overlayed';

axis1 label=('DXA') offset=(1 cm);

axis2 label=(angle=90);
proc gplot data=both ;

   symbol1 c=black i=join v=none height=2 repeat=&therepeat.;

   symbol2 c=red i=join v=none width=20;

  plot bmc*dxa=id/nolegend haxis=axis1 vaxis=axis2;

run; quit;

9. Plot by different categorical predictors. This will give mean lines (only), by group. Here I’m plotting by treatment group. You can cut and paste from step 9 and make underlined changes below (we are plotting means without individual profile plots).

proc sort data=long; by treatr dxa; run;

proc means data=long noprint nway;

var bmc;

by treatr dxa;

output out=mymeans mean=mean ;

where treatr ne .;

run;

goptions reset=all ftext=oldeng htext=3;

title 'Mean change in BMC over time by treatment randomization’;
axis1 label=('DXA') offset=(1 cm) minor=none ;

axis2 label=(angle=90)  minor=none;

proc gplot data=mymeans;

plot mean*dxa=treatr /haxis=axis1 vaxis=axis2  nolegend;

symbol1 v=dot i=join c=red r=1 line=1;

symbol2 v=dot i=join c=blue line=2;
run; quit;

10. It would be tedious to cut and paste the code from step 9 for each different predictor you’d like to graph. So, we can transform this code into a MACRO (function), for ease of future use. With a MACRO, you avoid: cutting and pasting; finding where you have to tweak the code each time; and generating excessively long code. You would like to make the MACRO as flexible as possible, and heavily commented (for later use).  Take your code from step 9 and change to match code below.

/*Variables:

Dataset: the dataset you would like to use (in long form)

Repeated: Your repeated outcome variable

Time: which number repeated measurement

ID: The variable that stores your id numbers/codes 

Predictor: Variable you would like to compare means by

Connector: interpolate options: join, spline, sm, rl, rc, rq, etc.

Font=font type 

*/
%macro meanplots (dataset=, repeated=, time=, id=, predictor=, connector=, font=); 

proc sort data=&dataset.; by &predictor. &time.; run;

proc means data=&dataset. noprint nway;

var &repeated.;
by &predictor. &time.;
output out=mymeans mean=mean ;

where &predictor. ne .;

run;

goptions reset=all ftext=&font. htext=3;

title “Mean change over time by &predictor.”;

axis1 label=('Time') offset=(1 cm) minor=none ;

axis2 label=(angle=90)  minor=none;

proc gplot data=mymeans;

plot mean*&time.=&predictor. /haxis=axis1 vaxis=axis2;

symbol1 v=dot i=&connector. c=red r=1 line=1;

symbol2 v=dot i=&connector. c=blue line=2;

symbol3 v=dot i=&connector. c=green line=3;

symbol4 v=dot i=&connector. c=black line=4;

symbol5 v=dot i=&connector. c=orange line=5;
run; quit;

%mend meanplots;

%meanplots (dataset=long, repeated=bmc, time=dxa, id=id, predictor=treatr, connector=join, font=titalic);

11. Now see how easy it is to run this with new predictors and to change font and plotting line!  Try other runs of the MACRO:

%meanplots (dataset=long, repeated=bmc, time=dxa, id=id, predictor=mencat1, connector=spline, font=oldeng);

%meanplots (dataset=long, repeated=bmc, time=dxa, id=id, predictor=sitenum, connector=sm60s, font=titalic);

12. We also have a continuous variable calc1, calcium at baseline; chop this into tertiles, and run the macro:
proc sort data=long; by calc1; run;
proc rank data=long out=ranks groups=3;

var calc1;

where calc1 ne .;

label calc1='tertile calcium';

run;
%meanplots (dataset=ranks, repeated=bmc, time=dxa, id=id, predictor=calc1, connector=sm60s, font=oldeng);

13. Strategy 1 to address statistical significance of calcium as a predictor: End point analysis (ANCOVA):

proc sort data=broad; by calc1; run;

proc rank data=broad out=ranksbroad groups=3;

var calc1;

where calc1 ne .;

label calc1='tertile calcium';

run;
proc glm data=ranksbroad;


class calc1;

    
model bmc3= bmc1 calc1 ;


lsmeans calc1 /pdiff adjust=tukey;

   run;

    The GLM Procedure

                                      Least Squares Means

                       Adjustment for Multiple Comparisons: Tukey-Kramer

                                                          LSMEAN

                               calc1     bmc3 LSMEAN      Number

                               0          2187.24758           1

                               1          2180.95664           2

                               2          2235.63492           3

                              Least Squares Means for effect calc1

                              Pr > |t| for H0: LSMean(i)=LSMean(j)

                                    Dependent Variable: bmc3

                         i/j              1             2             3

                            1                      0.9369        0.0267

                            2        0.9369                      0.0097

                        3        0.0267        0.0097
14. Ignoring calcium for the moment, let’s just explore whether there are significant changes in BMC over time. Start by running the naïve (incorrect!) ANOVA analysis—forgetting to take into account the correlation within subjects. 
proc anova data=long;


class dxa;

   
 model bmc = dxa ;


means dxa;

run;

Dependent Variable: bmc

                                              Sum of

      Source                      DF         Squares     Mean Square    F Value    Pr > F

      Model                        2        13827.88         6913.94       0.07    0.9315

      Error                      231     22499127.06        97398.82

      Corrected Total            233     22512954.94

      Source                      DF        Anova SS     Mean Square    F Value    Pr > F

  dxa                          2     13827.87519      6913.93759       0.07    0.9315

    MEANS:

Level of            -------------bmc-------------

                       dxa           N             Mean          Std Dev

                       1            78       2178.54744       309.334078

                       2            78       2190.05859       309.405180

                   3            78       2197.20808       317.454436

15. Now, run repeated-measures ANOVA, accounting for the variability that’s due to subject:

proc glm data=broad;

model bmc1-bmc3= / nouni; 

      repeated time;

run; quit;
16. Scroll through the output to find the folder “WITHIN” under repeated measures (univariate). Now the change looks significant, even after adjusting for violations of sphericity!

                                       The GLM Procedure

                             Repeated Measures Analysis of Variance

                   Univariate Tests of Hypotheses for Within Subject Effects

                                                                                    Adj Pr > F

Source                     DF    Type III SS    Mean Square   F Value   Pr > F    G - G    H - F

time                        2     13827.8752      6913.9376      3.87   0.0228   0.0313   0.0306
Error(time)               154    274833.4869      1784.6330


                               Greenhouse-Geisser Epsilon    0.8117

                           Huynh-Feldt Epsilon           0.8267

17. Scroll through to find the MANOVA output; gives generally same conclusion as univariate repeated measures: there is a significant increase over time.

  MANOVA Test Criteria and Exact F Statistics for the Hypothesis of no time Effect

                               H = Type III SSCP Matrix for time

                                     E = Error SSCP Matrix

        Statistic                        Value    F Value    Num DF    Den DF    Pr > F

       Wilks' Lambda               0.90688099       3.90         2        76    0.0244

       Pillai's Trace              0.09311901       3.90         2        76    0.0244

       Hotelling-Lawley Trace      0.10268052       3.90         2        76    0.0244

       Roy's Greatest Root         0.10268052       3.90         2        76    0.0244

18. We should also check the variance/covariance matrix for bmc1, bmc2, and bmc3 to assess whether compound symmetry is met (recall: compound symmetry is an assumption of univariate, but not multivariate repeated measures ANOVA).

proc corr data=broad cov;

var bmc1 bmc2 bmc3;

run;
Covariance Matrix, DF = 77

                                          bmc1              bmc2              bmc3

              bmc1                  95687.5715        94765.5011        95872.0531

              bmc2      BMC2        94765.5011        95731.5652        96205.0020

              bmc3      BMC3        95872.0531        96205.0020       100777.3187


                           Pearson Correlation Coefficients, N = 78

                                   Prob > |r| under H0: Rho=0

                                       bmc1          bmc2          bmc3

                         bmc1       1.00000       0.99014       0.97630

                                                   <.0001        <.0001

                         bmc2       0.99014       1.00000       0.97946

                         BMC2        <.0001                      <.0001

                         bmc3       0.97630       0.97946       1.00000

                 BMC3        <.0001        <.0001

19. Now, run repeated-measures ANOVA with a predictor: calcium tertile.
proc glm data=ranksbroad;

class calc1;

      model bmc1-bmc3= calc1/ nouni; 

      repeated time;

run; quit;

MANOVA output:

MANOVA Test Criteria and Exact F Statistics for the Hypothesis of no time Effect

                               H = Type III SSCP Matrix for time

                                     E = Error SSCP Matrix

                                      S=1    M=0    N=35.5

        Statistic                        Value    F Value    Num DF    Den DF    Pr > F

        Wilks' Lambda               0.88169748       4.90         2        73    0.0101
        Pillai's Trace              0.11830252       4.90         2        73    0.0101

        Hotelling-Lawley Trace      0.13417586       4.90         2        73    0.0101

        Roy's Greatest Root         0.13417586       4.90         2        73    0.0101

      MANOVA Test Criteria and F Approximations for the Hypothesis of no time*calc1 Effect

                            H = Type III SSCP Matrix for time*calc1

                                     E = Error SSCP Matrix

                                    S=2    M=-0.5    N=35.5

        Statistic                        Value    F Value    Num DF    Den DF    Pr > F

        Wilks' Lambda               0.85913629       2.88         4       146    0.0248
        Pillai's Trace              0.14109844       2.81         4       148    0.0277

        Hotelling-Lawley Trace      0.16368648       2.97         4    86.571    0.0237

    Roy's Greatest Root         0.16200001       5.99         2        74    0.0039

Univariate rANOVA output:
                                      The GLM Procedure

                             Repeated Measures Analysis of Variance

                        Tests of Hypotheses for Between Subjects Effects
      Source                      DF     Type III SS     Mean Square    F Value    Pr > F

      calc1                        2        53419.60        26709.80       0.09    0.9138

  Error                       74     21897978.52       295918.63

                              The GLM Procedure

                             Repeated Measures Analysis of Variance

                   Univariate Tests of Hypotheses for Within Subject Effects
                                                                                    Adj Pr > F

Source                     DF    Type III SS    Mean Square   F Value   Pr > F    G - G    H - F

time                        2     14932.3699      7466.1849      4.46   0.0131   0.0192   0.0177

time*calc1                  4     23549.3322      5887.3330      3.52   0.0089   0.0144   0.0130

Error(time)               148    247666.7626      1673.4241


                               Greenhouse-Geisser Epsilon    0.8215

                           Huynh-Feldt Epsilon           0.8604

20. Contrast these results with repeated-measures ANOVA comparing treatment group. Here, the graph indicates about a 20-gram BMC difference overall between the two groups and about a 20-gram increase over time, but not a difference in responses over time by group. Additionally, try out the polynomial option (recall, this assesses shape of the response over time: linear or quadratic are our only choices). Looks fairly linear.

proc glm data=broad;

class treatr;

      model bmc1-bmc3= treatr;

      repeated time 2 polynomial / summary;

run; quit;
MANOVA output:

MANOVA Test Criteria and Exact F Statistics for the Hypothesis of no time Effect

                               H = Type III SSCP Matrix for time

                                     E = Error SSCP Matrix

                                       S=1    M=0    N=36

        Statistic                        Value    F Value    Num DF    Den DF    Pr > F

        Wilks' Lambda               0.91055913       3.63         2        74    0.0312

        Pillai's Trace              0.08944087       3.63         2        74    0.0312

        Hotelling-Lawley Trace      0.09822631       3.63         2        74    0.0312

        Roy's Greatest Root         0.09822631       3.63         2        74    0.0312

    MANOVA Test Criteria and Exact F Statistics for the Hypothesis of no time*treatr Effect

                           H = Type III SSCP Matrix for time*treatr

                                     E = Error SSCP Matrix

                                      S=1    M=0    N=36

        Statistic                        Value    F Value    Num DF    Den DF    Pr > F


        Wilks' Lambda               0.99969151       0.01         2        74    0.9886

        Pillai's Trace              0.00030849       0.01         2        74    0.9886

        Hotelling-Lawley Trace      0.00030859       0.01         2        74    0.9886

    Roy's Greatest Root         0.00030859       0.01         2        74    0.9886

Univariate rANOVA output:

                                       The GLM Procedure

                             Repeated Measures Analysis of Variance

                        Tests of Hypotheses for Between Subjects Effects
      Source                      DF     Type III SS     Mean Square    F Value    Pr > F

      treatr                       1        25138.69        25138.69       0.09    0.7687

  Error                       75     21649911.77       288665.49

                                          17:21 Tuesday, May 10, 2005


  The GLM Procedure

                             Repeated Measures Analysis of Variance

                   Univariate Tests of Hypotheses for Within Subject Effects

                                                                                    Adj Pr > F

Source                     DF    Type III SS    Mean Square   F Value   Pr > F    G - G    H - F

time                        2     13183.1560      6591.5780      3.60   0.0298   0.0393   0.0378
time*treatr                 2        23.9062        11.9531      0.01   0.9935   0.9848   0.9866
Error(time)               150    274780.7579      1831.8717

                              Greenhouse-Geisser Epsilon    0.8116

                          Huynh-Feldt Epsilon           0.8380


Looking at shape of the response profile:

time_N represents the nth degree polynomial contrast for time

Contrast Variable: time_1

      Source                      DF     Type III SS     Mean Square    F Value    Pr > F

      Mean                         1      12936.9122      12936.9122       5.34    0.0236

      treatr                       1          0.0789          0.0789       0.00    0.9955

      Error                       75     181748.9791       2423.3197

Contrast Variable: time_2

      Source                      DF     Type III SS     Mean Square    F Value    Pr > F

      Mean                         1       246.24379       246.24379       0.20    0.6572

      treatr                       1        23.82726        23.82726       0.02    0.8901

  Error                       75     93031.77886      1240.42372

APPENDIX, LAB 5
Options for fonts in PROC GPLOT:

Syntax examples—to reset the default font for subsequent graphics: 

(ftext=font type, htext=font size):


goptions reset=all ftext=titalic htext=3;
[image: image9.png]Type Style Font Name Type Sample Uniform Fout
Brush BRUSH ABCaber23

Century

Bold CENTB ABCabc123 CENTBU
‘Bold Empty CENTBE ABCabcl28

Bold Tulic CENTBI ABCabc123 CENTBIU
Bold alic Empty CENTBIE ABCabel23

Expanded CENTX ABCabcl23 CENTXU
Expanded Empty CENTXE ABCabel28

Expanded ltalic CENTXI ABCabc12s CENTXIU
‘Expanded Ilic Empty  CENTXIE ABCabc123

German GERMAN ~ ABCabcl2al GERMANU
German Italic GITALIC HBGobcr23 GITALICU
Hershey

Sans Serif SIMPLEX ABCabc123 SIMPLEXU
Sans Serif Bold DUPLEX ABCabc123 DUPLEXU
Serif COMPLEX ABCabc123 COMPLEXU
Serif Bold ‘TRIPLEX ABCabc123 TRIPLEXU
‘Serif Bold Ttalic: TITALIC ABCabe123 TITALICU
Serif Talic mALIC ABCabc123 TTALICU
01d English OLDENG ABGabrl23 OLDENGU
Seript SCRIPT ABCator23

Cseript CSCRIPT ABEabel23

Swiss SWISS. ABCabc123 SWISSU
Empty SWISSE ABCabci23

Bold SWISSB. ABCabc123 SWISSBU
‘Bold Empty SWISSBE ABCeabei23

Bold Italic SWISSBI ABCabc123 SWISSBIU

(continued)





[image: image10.png]Type Style

‘Bold Talic Empty
Expended
Expanded Empty
‘Expanded Bold
Expanded Bold Empty
Talie
Halic Empty
Light
Light Empty

Zapt
Empty
Bold
Bold Empty
Bold Ialic
Bold Italic Empty
Ialic
Italc Empty

Font Name

ZAPFBIE

2ZAPFIE

Type Sample.

ABCebei123
ABCabc123
ABCabci23
ABCabc123
ABGCabci28
ABCabc123
ABCabci23
ABCabci23
ABCabc123
ABCabc123
ABCabc123
ABCabc123
ABCabc123
ABCabc123
ABCabci123
ABCabc123
ABCabc123

SWISSXU

SWISSXBU

SWISSIU

SWISSLU

ZAPFBU

ZAPFBIU

ZAPFIU





If you want reinforcement on repeated measures ANOVA, try this website (online seminar from UCLA covering many of the same topics as I did on Monday):

http://www.ats.ucla.edu/stat/sas/seminars/sas_repeatedmeasures/default.htm

DXA*calcium is significant, Overall, changes over time are significant.














time*calcium is significant, indicating a difference between at least two of the calcium groups in their response profile over time, as we saw in the graph.





But response profile does not differ by group.





Dataset work.ranks will contain a variable calc1 that is ranks from 0-2.





Y-axis range= 120  grams





Linear is clearly the better choice for shape.





Changes over time differ between calcium groups.





Same conclusions as MANOVA!





No  significant difference overall between groups.





Response profiles over time are not different between groups.





Increase over time is statistically significant.





No overall differences between calcium groups.





Changes over time are significant.





Increase over time is statistically significant.





-Under compound symmetry, we estimate 1 variance and 1 covariance (cost is only 2 degrees of freedom).


-MANOVA allows a completely “unstructured” variance/covariance matrix, where all 6 variances and covariances are estimated (costs 6 degrees of freedom).


-We’ll discuss alternatives to  compound symmetry and unstructured next week.





-Variance looks just slightly higher for bmc3.


-The three covariances/correlations look similar with the highest correlation between bmc1 and bmc2.


-Looks like compound symmetry is relatively well satisfied.








Notice the ENORMOUS reduction in unexplained variability; but variability due to time (SSB for time) does not change.





Sphericity is not too badly violated….





No evidence of change over time!





Highest tertile of calcium intake looks different from the lower tertiles at the end of the study, even after adjusting for baseline differences.





Now, that looks interesting!  Looks like the three groups are different in their responses over time (highest tertile of calcium intake does well. But is this statistically significant?? IMPORTANT to consider the scale of the Y-axis!! (here Y-axis range= 70 grams)





Set global font type (ftext) and font size (htext) for graphics. Here, I’m choosing Old English font. See Appendix for other font types.





We do not want the means calculated for individuals missing a value of the predictor.





Offset will offset your plot from the specified axis.





You must sort before plotting, or lines will be connected incorrectly. 





Since we know that the means will be plotted last (id=1000), we call 78 thin black lines first and then one thick red line over the other lines. We know we have therepeat number of individuals in our datset.  Global variables are sandwiched by an ampersand and a period.





_Freq_ is automatically generated. =the number of observations included in the average. 





The dataset Work.means:





This stores _freq_ as a global variable, which I’m calling therepeat. We can call this value later for graphing.





Only the means have a _freq_ value,  so I can flag the mean observations by giving them an id number of 1000 (which will be sorted to end of dataset).





Code for concatenating two datasets.  Means will be first three records.




















nway omits calculation of an overall mean.


noprint suppresses printing means to SAS output window.





Note: SAS generates Y-axis automatically, range of 50 grams





Y-axis range= 150  grams





Saves means in dataset work.means.





By not specifying a variable name to contain the means, the means will be called the original variable name, bmc, which will facilitate graphing.





Plot bone mineral content by dxa measurement by person.  





Y-axis range= 50 grams





Asks for tertiles.





No legend is added to the graph because the label for sitenum (set previously) is too long!





Must specify one symbol/line for each individual in the dataset. Here there are 78 individuals. Repeat means that you want the first 78 symbols used to be identical.  











Outputs 1 record for each bmc1; each bmc2; and each bmc3.





The keep statement keeps ONLY the variables specified.





When used BEFORE the set statement, the retain statement allows you to specify the order of variables in your dataset (column order). “Retain” is misleading, since non-specified variables will still be retained at the end of the dataset.

